

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

AFS DETECTION LIMITS

**Listing and explanation of a computer program
for the Baird Plasma/AFS and Apple II computer
to determine detection limits**

By

Allen L. Meier and Robert C. Bigelow

Open-File Report 84-698

This report has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the U.S. Geological Survey.

1984

Contents

	Page
Abstract.....	1
Introduction.....	1
Program description.....	2
Program execution.....	4
A Computer listing of the AFS DETECTION LIMITS program.....	10
Program Output Section.....	20
Detection limits program example.....	21

Abstract

AFS DETECTION LIMITS is a computer program written specifically for an Apple II PLUS computer with 48K of memory, one disk drive, and an Epson MX-80 printer. The computer is connected to a PLASMA/AFS Atomic Fluorescence Spectrometer manufactured by Baird Corporation. The program is menu driven and prompts the operator for all input. The program collects raw signal data from the PLASMA/AFS and then calculates, displays and prints the mean, standard deviation, and relative standard deviation of the signal data, and detection limit for each element determined in a standard solution. These statistics and detection limits are used to check instrument performance and to set up the operating conditions of the instrument. A data collection routine is also provided in the AFS DETECTION LIMITS program to collect and print data from the menu modes of the PLASMA/AFS instrument.

Introduction

The Atomic Fluorescence Spectrometer (PLASMA/AFS) manufactured by Baird Corporation is capable of determining from 1 to 12 elements simultaneously. The instrument has two modes of operation: (1) the menu mode allows for the calibration of each of the twelve channels of the instrument and the printout of results on an internal printer; and (2) the command mode provides raw signal data but has no provision for printing this raw data. The raw signal data are used to set up the instrument and calculate detection limits of elements in standard solutions. Detection limits are used to check for proper instrument performance and to determine the best instrumental parameters and plasma conditions to use. The detection limit is defined, in this report, as the concentration of an element that will produce a signal that is twice the standard deviation of ten repetitions of the signal produced by a blank.

The Baird PLASMA/AFS generates data rapidly but has no provision for printing raw data or calculating detection limits using the internal computer of the instrument. Because the raw signal data and detection limit information are used to set up the instrument, an external computer is useful, if not essential, for these data collections and calculations. By connecting an Apple II PLUS Computer to the external printer interface of the Baird PLASMA/AFS and using the program described in this report, detection limits and raw instrumental data can be collected and printed.

AFS DETECTION LIMITS is a computer program written specifically for an Apple II PLUS Computer with 48K of memory, one disk drive, and an Epson MX-80 printer. It is to be used to collect instrumental data from a Baird PLASMA/AFS and calculate detection limits of elements in standard solutions and has no other intended purpose. The program is menu driven and prompts the operator for all input. The program collects raw signal data from the Baird PLASMA/AFS, and then calculates, displays and prints the mean, standard deviation, relative standard deviation, and detection limit for each element in a standard solution. A data collection routine is also provided in the AFS DETECTION LIMITS program to collect and print data from the menu modes of the Baird PLASMA/AFS instrument. The user of this program must have a knowledge of the Baird PLASMA/AFS and its operation, however, little computer experience is necessary to use the program.

Program description

This program is written in Applesoft Basic Language for an Apple II PLUS computer system with 48K memory and one disk drive and an Epson MX-80 printer. The program is menu driven and prompts the operator for all input. All inputs use "Y" for yes, "N" for no, and "Return" for a default or no change. Other inputs are entered by pressing a numbered key, or if operator

input information is requested, by typing numbers or words followed by the "Return" key.

The program consists of three sections: input; data collection and reduction; and output.

The input section requires that the operator enter information in response to prompting menus, as they appear on the screen. The information entered is used in the data collection and reduction and output sections for labeling and heading information and for selecting data for manipulation using output routines provided. These "Header" conditions are stored on disk and are retrieved by the program; therefore, only minor changes are usually required in subsequent runs.

The data collection and reduction section of the program passes control of the computer to the Baird PLASMA/AFS instrument for data input. Data, for one to 12 elements, are transmitted from the test exerciser mode of the Baird PLASMA/AFS instrument to the Apple II computer. A maximum of 10 repetitions may be made on a maximum of 10 runs of a standard solution. Between data collection repetitions provision is made for the operator to abort the data collection and to reject data before they are used in the calculations performed by the program. The mean, standard deviation, relative standard deviation, and detection limit are calculated for each element in each standard solution run. The detection limit is calculated using the mean and standard deviation of the last blank run and the standards run by the formula:

$$D = \frac{2VC}{S-B}$$

Where, D is the detection limit, V is the standard deviation of the blank signal, C is the concentration of the standard, S is the mean standard signal, and B is the mean blank signal. The blank values used in the calculations are always the last ones run. The data collection routine displays or prints data from the menu modes of the PLASMA/AFS instrument.

The output section provides for display and/or printout of the data collected and calculated. Printer commands for the Epson MX-80 printer are embedded in the program to provide a readable and useful printout. Options are provided to allow the operator to print as much or as little of the header information as is needed to indicate the conditions under which the data were collected.

Program execution

The following are the prompts that appear on the computer screen, shown as underlined characters. A brief description of each prompt is given with the possible responses. The numbers in parentheses are the program line numbers at which these prompts occur.

HOW MANY ELEMENTS ARE YOU RUNNING? (210)

A number from 0 to 12 is entered and the return key pressed. If 0 is entered, the program goes directly to the ending routine. If a number greater than 12 is entered, the prompt will return.

SELECT PRINTER OPTION (240)

0= NOTHING

1= EVERYTHING

ENTER OPTION NUMBER

Only 0 or 1 will be accepted. If 0 is entered, nothing is printed. If 1 is entered, all data requested in the next menu are printed. The next menu appears as soon as a 0 or 1 is entered.

SELECT DATA OPTION (320)

0= ALL DATA FULL HEADER

1= ALL DATA ELEMENT INFO.

2= ALL DATA NO HEADER

3= DETECTION LIMITS FULL HEADER

4= DETECTION LIMITS ELEMENT INFO.

5= DETECTION LIMITS NO HEADER

6= DATA FROM KEYBOARD FULL HEADER

7= DATA FROM KEYBOARD ELEMENT INFO.

8= DATA FROM KEYBOARD NO HEADER

9= PRINT FROM MENU MODE

ENTER OPTION NUMBER (440)

When 0, 3, or 6 (full header) is selected, all information contained in the header file will be displayed or printed next to the titles shown in the next menu (title, operator, day, month, year, time, power, carrier flow, coolant flow, pump speed, remarks, remarks, remarks, and remarks). Likewise, if 1, 4, or 7 (element information) is chosen, that is all that will be displayed or printed under the titles of Position, Element, Current, and PMT. Entry of 6, 7, or 8 allows the operator to input data manually. Selection 9 provides the use of the computer and printer as an external printer from the menu modes of the instrument.

TITLE

OPERATOR

DAY

MONTH

YEAR

TIME

POWER

CARRIER FLOW

COOLANT FLOW

PUMP SPEED

REMARKS

REMARKS

REMARKS

REMARKS

DO YOU WANT TO CHANGE ANYTHING? (1700)

If a header was selected in the previous menu, any information contained in the header disk file will be displayed on the screen with each of the above titles (title, operator, day, month, year, time, power, carrier flow, coolant flow, pump speed, remarks, remarks, remarks, and remarks). If any key except "Y" is pressed, all of the information is retained and used in subsequent printouts. If "Y" is pressed, the following prompt is displayed on the screen, and each of the above titles (title, operator, day, month, year, time, power, carrier flow, coolant flow, pump speed, remarks, remarks, remarks, and remarks) is displayed one at a time with its current contents in inverse video.

RETURN FOR NO CHANGE < / > TO DELETE (1860)

As the prompt indicates, pressing the return key retains the contents of that section of the header as displayed in inverse video. If a "/" key is entered, that section is deleted and nothing will be displayed or printed in that position of the header. Anything typed into the keyboard and followed by the return key replaces the contents of that section. After a response is made to the last remark, the header, with changes, is once again displayed and can be accepted or further changed.

<u>POSITION</u>	<u>ELEMENT</u>	<u>CURRENT</u>	<u>PMT</u>
-----------------	----------------	----------------	------------

DO YOU WANT TO CHANGE ANYTHING? (1860)

The contents of the element information file will be displayed under each of the headings (position, element, current, pmt) for the number of elements being run. If any key except "Y" is pressed, all of the above information is retained and used in subsequent printouts.

If "Y" is entered, the following message is displayed, and each of the above titles (position, element, current, pmt) is displayed, one at a time, with its contents in inverse video.

RETURN FOR NO CHANGE < / > TO DELETE (1860)

As the prompt indicates, pressing the return key retains the contents of that section of the header as displayed in inverse video. If a "/" key is entered, that section is deleted and nothing will be displayed or printed in that position of the header. Anything typed into the keyboard and followed by the return key replaces the contents of that section. After responding to the last PMT (photomultiplier tube voltage), the header, as changed, is once again displayed and can be accepted or altered. The positions of the elements can be entered in any order but they must be in the range of 1 to 12 and be the module positions in use by the AFS instrument. The program will not accept two positions with the same number or two elements with the same symbol.

HOW MANY REPETITIONS DO YOU WANT? (2130)

The number of repeats desired is entered. The range must be from 0 to 10 or an error message will be displayed and the question will be asked again. If 0 is entered, the program will go to the end of data input section and ask if more standards are to be run. If a number between 1 and 10 is entered, the program continues.

ENTER REMARK FOR THIS RUN (2160)

A remark of up to 255 characters can be entered for the run. The return key is pressed, if no remark is desired.

FOR RUN " "

IS CONCENTRATION = FOR ALL ELEMENTS (2160)

The return key or "Y" is pressed if the standard to be run contains all of the elements at the same concentration. The prompt below (WHAT IS THE

CONCENTRATION IN G/ML?) appears. If the standard to be run contains the elements at different concentrations, then "N" is entered and the second prompt (ENTER CONCENTRATION IN G/ML FOR ELEMENT " ") appears.

WHAT IS THE CONCENTRATION IN G/ML: (2170)

A concentration value that equals the standard is entered. If the solution is a blank, then 0 is entered.

ENTER CONCENTRATION IN G/ML FOR ELEMENT " " (2190)

This prompt will appear for each element being run and requires that a concentration be entered for each.

DO YOU HAVE ANY CHANGES? (2220)

If "Y" is entered, the program returns to the prompt, "HOW MANY REPETITIONS DO YOU WANT?", and all entries must be re-entered. If the return key is pressed, all entries are accepted and the program continues by going directly to accepting data from the instrument or keyboard. The instrument must be in the test exerciser mode with the external print command given. Control of the computer is taken away from the operator until data are transferred from the instrument to the computer. At that time the following message appears and "*" are slowly printed across the screen.

HIT CTRL C TO ABORT: RUN " " OF " " (2310)

The run and the data taken in the run can be canceled by pressing the CTRL and C keys simultaneously, while the "*" are being printed across the screen. The program then returns to the prompt, "HOW MANY REPETITIONS DO YOU WANT?", and the run can be restarted. If the data collection is not aborted, the data are collected for the number of repeats selected. The computer will beep and the following prompt will appear.

RUN " " OF " " DONE CHANGE STANDARDS! (2300)

The computer will take some time making calculations before printing the newly calculated data on the screen. Then the following prompt will be displayed.

DO YOU HAVE MORE STANDARDS TO RUN? (2640)

If the return key is pressed, the program will ask for the number of repetitions for the next run and go through the same data collection routine. If "N" is entered, the data selected will be printed on the printer. If no printout was selected, the data will be printed on the screen. The following prompt will be displayed after the print.

DO YOU WANT TO CONTINUE USING THE SAME ELEMENTS AND SAME DATA? (3130)

If "Y" is entered, then the next prompt appears. If the return key is pressed, the second prompt appears.

DO YOU WANT TO CHANGE PRINT INSTRUCTIONS? (3140)

If "Y" is entered, then the program returns to the beginning, prompts for changes, except the number of elements which cannot be changed. If any other key is entered, the program returns to the prompt, "DO YOU HAVE MORE STANDARDS TO RUN?".

DO YOU WANT TO RUN NEW DATA? (3160)

If "Y" is entered, the program starts from the beginning. If any other key is entered, the next prompt appears.

DO YOU WANT TO QUIT? (3190)

If any key other than "Y" is pressed, the program returns to the ending prompts starting at "DO YOU WANT TO CONTINUE USING THE SAME ELEMENTS AND SAME DATA?". If "Y" is entered, the following prompt appears.

ATOMIC FLUORESCENCE PROGRAM ENDED! ENTER RUN TO BEGIN AGAIN (3200)

A Computer listing of the AFS DETECTION LIMITS program

```

100 REM THIS PROGRAM COLLECTS DATA FROM THE BAIRD ATOMIC
FLUORESCENCE ICP CALCULATES STATISTICS, AND DETECTION LIMITS.
110 REM PROVISION IS MADE TO USE THE COMPUTER AND PRINTER AS AN
EXTERNAL PRINTER
120 REM LAST UPDATED ON 11 JANUARY 1984
130 REM WRITTEN BY ALLEN MEIER AFTER A PROGRAM OF ROBERT BIGELOW
OF USGS DENVER, CO
140 REM *****
150 REM **PRINTER MX-80 EPSON**
160 REM *****
170 DIM Z$(12), B$(12), C$(12), D$(12,10), A$(100),
EL(12,10,10), RS(12,10), SQ(12,10), ST(12,10), SU(12,10), EL$(12),
LC$(12), PM$(12), MP$(12), D(12,10), ME(12,10), C(12,10),
RS$(12,10), ST$(12,10), ME$(12,10), ES(12)
180 D$ = CHR$(4)
190 HOME
200 RESTORE
210 INPUT "HOW MANY ELEMENTS ARE YOU RUNNING?"; NE: IF NE = 0
GOTO 3190
220 DATA "A","B","C","D","E","F","G","H","I","J","K","L"
230 FOR N = 1 TO NE: READ NE$: NEXT
240 HOME : INVERSE : PRINT "SELECT PRINTER OPTION": NORMAL :
PRINT
250 ONERR GOTO 240
260 PRINT : PRINT "0= NOTHING "
270 PRINT : PRINT "1= EVERYTHING "
280 PRINT : PRINT : INVERSE : PRINT "ENTER OPTION NUMBER": NORMAL
290 GET A: POKE 216,0
300 A = A + 1
310 IF A > 3 GOTO 240
320 HOME : INVERSE : PRINT "SELECT DATA OPTION": NORMAL : PRINT
330 ONERR GOTO 320
340 PRINT : PRINT "0= ALL DATA FULL HEADER"
350 PRINT : PRINT "1= ALL DATA ELEMENT INFO."
360 PRINT : PRINT "2= ALL DATA NO HEADER"
370 PRINT : PRINT "3= DETECTION LIMITS FULL HEADER"
380 PRINT : PRINT "4= DETECTION LIMITS ELEMENT INFO."
390 PRINT : PRINT "5= DETECTION LIMITS NO HEADER"
400 PRINT : PRINT "6= DATA FROM KEYBOARD FULL HEADER"
410 PRINT : PRINT "7= DATA FROM KEYBOARD ELEMENT INFO."
420 PRINT : PRINT "8= DATA FROM KEYBOARD NO HEADER"
430 PRINT : PRINT "9= PRINT FROM MENU MODE"
440 PRINT : PRINT : INVERSE : PRINT "ENTER OPTION NUMBER": NORMAL
450 GET H: POKE 216,0
460 H = H + 1
470 IF H > 10 GOTO 240
480 ON H GOTO 490,550,550,490,550,550,490,550,550,3710
490 ONERR GOTO 530
500 PRINT D$: PRINT D$;"OPEN HEADFILE"
510 PRINT D$;"READ HEADFILE"

```

```

520 INPUT TI$, OP$, DA$, MO$, YR$, T$, POS$, CA$, CO$, RE$, OX$,
    PU$, R1$, R2$, R3$, R4$
530 PRINT D$;"CLOSE HEADFILE"
540 POKE 216,0
550 ONERR GOTO 610
560 PRINT D$: PRINT D$;"OPEN";NE$
570 PRINT D$;"READ";NE$
580 INPUT NE
590 FOR I = 1 TO NE: INPUT EL$(I),MP$(I),LC$(I),PM$(I)
600 NEXT I PRINT D$;"CLOSE";NE$
620 POKE 216,0
630 GOTO 1490
640 HOME : ON H GOTO 680, 1070, 1070, 680, 1070, 1070, 680, 1070,
    1070, 3710
650 REM *****
660 REM *HEADER INPUT*
670 REM *****
680 PRINT "TITLE ";
690 INVERSE : PRINT TI$: NORMAL : INPUT X$: IF X$ < > "" THEN
    TI$ = X$: IF X$ = "/" THEN TI$ = ""
700 PRINT "OPERATOR ";
710 INVERSE : PRINT OP$: NORMAL : INPUT X$: IF X$ < > "" THEN
    OP$ = X$: IF X$ = "/" THEN OP$ = ""
720 PRINT "DAY ";
730 INVERSE : PRINT DA$: NORMAL : INPUT X$: IF X$ < > "" THEN
    DA$ = X$: IF X$ = "/" THEN DA$ = ""
740 PRINT "MONTH ";
750 INVERSE : PRINT MO$: NORMAL : INPUT X$: IF X$ < > "" THEN
    MO$ = X$: IF X$ = "/" THEN MO$ = ""
760 PRINT "YEAR ";
770 INVERSE : PRINT YR$: NORMAL : INPUT X$: IF X$ < > "" THEN
    YR$ = X$: IF X$ = "/" THEN YR$ = ""
780 PRINT "TIME ";
790 INVERSE : PRINT T$: NORMAL : INPUT X$: IF X$ < > "" THEN T$
    = X$: IF X$ = "/" THEN T$ = ""
800 PRINT "POWER ";
810 INVERSE : PRINT POS$: NORMAL : INPUT X$: IF X$ < > "" THEN
    POS$ = X$: IF X$ = "/" THEN POS$ = ""
820 PRINT "CARRIER FLOW ";
830 INVERSE : PRINT CA$: NORMAL : INPUT X$: IF X$ < > "" THEN
    CA$ = X$: IF X$ = "/" THEN CA$ = ""
840 PRINT "COOLANT FLOW ";
850 INVERSE : PRINT CO$: NORMAL : INPUT X$: IF X$ < > "" THEN
    CO$ = X$: IF X$ = "/" THEN CO$ = ""
860 PRINT "REDUCTANT FLOW ";
870 INVERSE : PRINT RE$: NORMAL : INPUT X$: IF X$ < > "" THEN
    RE$ = X$: IF X$ = "/" THEN RE$ = ""
880 PRINT "OXIDANT FLOW ";
890 INVERSE : PRINT OX$: NORMAL : INPUT X$: IF X$ < > "" THEN
    OX$ = X$: IF X$ = "/" THEN OX$ = ""
900 PRINT "PUMP SPEED ";
910 INVERSE : PRINT PU$: NORMAL : INPUT X$: IF X$ < > "" THEN
    PU$ = X$: IF X$ = "/" THEN PU$ = ""
920 PRINT "REMARKS ";

```

```

930 INVERSE : PRINT R1$: NORMAL : INPUT X$: IF X$ < > "" THEN
R1$ = X$: IF X$ = "/" THEN R1$ = ""
940 PRINT "REMARKS ";
950 INVERSE : PRINT R2$: NORMAL : INPUT X$: IF X$ < > "" THEN
R2$ = X$: IF X$ = "/" THEN R2$ = ""
960 PRINT "REMARKS ";
970 INVERSE : PRINT R3$: NORMAL : INPUT X$: IF X$ < > "" THEN
R3$ = X$: IF X$ = "/" THEN R3$ = ""
980 PRINT "REMARKS ";
990 INVERSE : PRINT R4$: NORMAL : INPUT X$: IF X$ < > "" THEN
R4$ = X$: IF X$ = "/" THEN R4$ = ""
1000 REM *****
1010 REM *END HEADER INPUT*
1020 REM *****
1030 POKE 34,0: HOME : GOTO 1550
1040 REM *****
1050 REM *ELEMENT INFOR INPUT*
1060 REM *****
1070 IF NE = 0 THEN GOTO 1200
1080 FOR I = 1 TO NE
1090 PRINT "MODULE POSITION ";
1100 INVERSE : PRINT MP$(I): NORMAL : INPUT X$: IF X$ < > ""
THEN MP$(I) = X$: IF X$ = "/" THEN MP$(I) = ""
1110 IF VAL (MP$(I)) > 12 THEN FLASH : GOTO 1090
1120 IF VAL (MP$(I)) < = 0 THEN FLASH : GOTO 1090
1130 PRINT "ELEMENT ";
1140 INVERSE : PRINT EL$(I): NORMAL : INPUT X$: IF X$ < > ""
THEN EL$(I) = X$: IF X$ = "/" THEN EL$(I) = ""
1150 PRINT "LAMP CURRENT ";
1160 INVERSE : PRINT LC$(I): NORMAL : INPUT X$: IF X$ < > ""
THEN LC$(I) = X$: IF X$ = "/" THEN LC$(I) = ""
1170 PRINT "PHOTOMULTIPLIER SETTING ";
1180 INVERSE : PRINT PM$(I): NORMAL : INPUT X$: IF X$ < > ""
THEN PM$(I) = X$: IF X$ = "/" THEN PM$(I) = ""
1190 NEXT
1200 HOME
1210 FOR I = 1 TO NE
1220 FOR X = 1 TO NE
1230 IF I = X THEN GOTO 1260
1240 IF MP$(I) = MP$(X) THEN POKE 34,0: HOME : FLASH : PRINT "
MODULE POSITION ";MP$(I);" USED TWICE!": NORMAL : GOTO 1750
1250 IF EL$(I) = EL$(X) THEN POKE 34,0: HOME : FLASH : PRINT "
ELEMENT ";EL$(I);" USED TWICE!": NORMAL : GOTO 1750
1260 NEXT
1270 NEXT
1280 FOR I = 1 TO 12
1290 Z = VAL (MP$(I))
1300 Z$(Z) = MP$(I)
1310 B$(Z) = EL$(I)
1320 C$(Z) = LC$(I)
1330 E$(Z) = PM$(I)
1340 NEXT
1350 X = 1
1360 FOR I = 1 TO 12
1370 IF Z$(I) = "" THEN GOTO 1440

```

```

1380 MP$(X) = Z$(I)
1390 EL$(X) = B$(I)
1400 LC$(X) = C$(I)
1410 PM$(X) = E$(I)
1420 X = X + 1
1430 Z$(I) = ""
1440 NEXT
1450 POKE 34,0: HOME
1460 REM *****
1470 REM END ELEMENT INFO INPUT*
1480 REM *****
1490 HOME : ON H GOSUB 1550, 1750, 1750, 1550, 1750, 1750, 1550,
1750, 1750, 3710
1500 IF BB < > 1 GOTO 2120
1510 GOTO 1910
1520 REM *****
1530 REM *HEADER OUTPUT*
1540 REM *****
1550 IF TIS < > "" THEN PRINT "TITLE ";TIS
1560 IF OPS < > "" THEN PRINT "OPERATOR ";OPS
1570 IF (DAS < > "") AND (MOS < > "") AND (YRS < > "") THEN
PRINT "DATE ";DAS;" ";MOS;" ";YRS;
1580 IF TS < > "" THEN PRINT "      TIME ";TS
1590 IF TS = "" THEN PRINT
1600 IF POS < > "" THEN PRINT "POWER ";POS
1610 IF CAS < > "" THEN PRINT "CARRIER FLOW ";CAS
1620 IF COS < > "" THEN PRINT "COOLANT FLOW ";COS
1630 IF RES < > "" THEN PRINT "REDUCTANT FLOW ";RES
1640 IF OXS < > "" THEN PRINT "OXIDANT FLOW ";OXS
1650 IF PUS < > "" THEN PRINT "PUMP SPEED ";PUS
1660 IF R1S < > "" THEN PRINT "REMARKS ";R1S
1670 IF R2S < > "" THEN PRINT "REMARKS ";R2S
1680 IF R3S < > "" THEN PRINT "REMARKS ";R3S
1690 IF R4S < > "" THEN PRINT "REMARKS ";R4S
1700 INPUT "DO YOU WANT TO CHANGE ANYTHING?";BS: IF BS = "Y" THEN
BB = 1: INVERSE : PRINT "RETURN FOR NO CHANGE < / > TO DELETE":
NORMAL : PRINT : POKE 34, PEEK (37): GOTO 640
1710 REM *****
1720 REM *HEADER OUT END*
1730 REM *****
1740 POKE 34,0: HOME
1750 PRINT "POSITION  ELEMENT  CURRENT  PMT"
1760 REM *****
1770 REM **ELEMENT OUT**
1780 REM *****
1790 FOR I = 1 TO NE
1800 HTAB 4: PRINT MP$(I);
1810 HTAB 13: PRINT EL$(I);
1820 HTAB 22: PRINT LC$(I);
1830 HTAB 30: PRINT PM$(I)
1840 NEXT
1850 P = FRE (0)
1860 INPUT "DO YOU WANT TO CHANGE ANYTHING?";BS: IF BS = "Y" THEN
BB = 1: INVERSE : PRINT "RETURN FOR NO CHANGE < / > TO DELETE":
NORMAL : PRINT : POKE 34, PEEK (37): GOTO 1070

```

```

1870 RETURN
1880 REM *****
1890 REM **DISK HEADER FILES**
1900 REM *****
1910 ON H GOTO 1920,1980,1980,1920,1980,1980,1920,1980,1980,3710
1920 PRINT D$;"OPEN HEADFILE"
1930 PRINT D$;"DELETE HEADFILE"
1940 PRINT D$;"OPEN HEADFILE"
1950 PRINT D$;"WRITE HEADFILE"
1960 PRINT TI$: PRINT OP$: PRINT DA$: PRINT MO$: PRINT YR$: PRINT
T$: PRINT PO$: PRINT CA$: PRINT CO$: PRINT RE$: PRINT OX$: PRINT
PU$: PRINT R1$: PRINT R2$: PRINT R3$: PRINT R4$
1970 PRINT D$;"CLOSE HEADFILE"
1980 IF BB < > 1 GOTO 2120
1990 PRINT D$;"OPEN";NE$
2000 PRINT D$;"DELETE";NE$
2010 PRINT D$;"OPEN";NE$
2020 PRINT D$;"WRITE";NE$
2030 PRINT NE
2040 FOR I = 1 TO NE: PRINT EL$(I): PRINT MP$(I): PRINT LC$(I):
PRINT PM$(I)
2050 NEXT
2060 PRINT D$;"CLOSE";NE$
2070 BB = 0
2080 REM *****
2090 REM **DISK HEADER END**
2100 REM **STAT. INITIALIZATION **
2110 REM *****
2120 RN = RT + 1:RT = RT + 1
2130 SPEED= 255: PRINT : PRINT : INPUT "HOW MANY REPETITIONS DO
YOU WANT? ";R(RN): IF R(RN) = 0 THEN RN = RT - 1:RT = RT - 1:
GOTO 2640
2140 IF R(RN) > 10 THEN INVERSE : PRINT "REPETITIONS MUST BE 10
OR LESS.": NORMAL : GOTO 2130
2150 PRINT : PRINT "ENTER REMARK FOR THIS RUN": INPUT R$(RN)
2160 PRINT : HTAB 14: PRINT "FOR RUN ";RN: PRINT "IS
CONCENTRATION = FOR ALL ELEMENTS?": GET B$
2170 PRINT : PRINT : PRINT : IF B$ < > "N" THEN INPUT "WHAT IS
THE CONCENTRATION IN UG/ML?";CC: FOR I = 1 TO NE:C(I,RN) = CC:
NEXT : GOTO 2220
2180 FOR I = 1 TO NE
2190 PRINT : PRINT "ENTER CONCENTRATION IN UG/ML FOR ELEMENT
";EL$(I);
2200 INPUT C(I,RN)
2210 NEXT
2220 PRINT : PRINT "DO YOU HAVE ANY CHANGES?": GET B$: IF B$ =
"Y" GOTO 2130
2230 ON H GOTO 2240,2240,2240,2240,2240,2240,3260,3260,3260,3710
2240 ONERR GOTO 2130
2250 FOR I = 1 TO R(RN)
2260 REM *****
2270 REM **INSTRUMENT INPUT**
2280 REM *****
2290 PRINT D$: PRINT D$;"IN#2": INPUT "";A$(I): PRINT D$;"IN#0":
IF A$(I) = "" THEN I = I - 1

```



```

2300 IF I = R(RN) THEN PRINT CHR$(7), CHR$(7): PRINT :
INVERSE : PRINT "RUN ";I;" OF ";R(RN);" DONE "; "CHANGE
STANDARDS!": NORMAL : GOTO 2330
2310 PRINT "HIT CTRL C TO ABORT! "; "RUN ";I;" OF ";R(RN)
2320 FOR Z = 0 TO 79: SPEED= 10: PRINT "*";: NEXT : SPEED= 255
2330 IF LEN (A$(I)) < 10 * NE THEN A$(I) = " " + A$(I): GOTO
2330
2340 FOR N = 1 TO NE
2341 L = ((N - 1) * 10) + 1
2343 FOR J = L TO L + 9
2344 IF ( MID$( A$(I),J,1) = "-" ) THEN SIGN = - 1
2350 NEXT
2360 REM *****
2370 REM **STATISTICS CALCULATION**
2380 REM *****
2390 EL(N,RN,I) = VAL ( MID$( A$(I),L,10)) * SIGN
2400 SUM(N,RN) = SUM(N,RN) + EL(N,RN,I)
2401 SQ(N,RN) = SQ(N,RN) + EL(N,RN,I) ^ 2
2402 NEXT
2403 NEXT
2404 P = FRE (0)
2405 POKE 216,0
2410 FOR N = 1 TO NE
2420 MEAN(N,RN) = SU(N,RN) / R(RN)
2430 IF R(RN) - 1 = 0 GOTO 2470
2440 IF ((SQ(N,RN) - ((SU(N,RN) ^ 2) / R(RN))) / (R(RN) - 1)) <
= 0 GOTO 2470
2450 STDV(N,RN) = SQR ((SQ(N,RN) - ((SU(N,RN) ^ 2) / R(RN))) /
(R(RN) - 1))
2460 RSD(N,RN) = ST(N,RN) * 100 / ME(N,RN)
2470 PRINT "ELEMENT ";EL$(N); " RUN NUMBER= ";RN;" REPETITIONS=
";R(RN); " CONC.= "C(N,RN); " "R$(RN)
2475 ST%(N,RN) = ST(N,RN)
2476 RS$(N,RN) = STR$( RS(N,RN))
2480 IF ME(N,RN) > 32767 THEN PRINT "MEAN= ";ME(N,RN); " STD.
DEV.= ";ST%(N,RN); " RSD%= "; LEFT$( RS$(N,RN),4): GOTO 2510
2490 ME%(N,RN) = ME(N,RN)
2500 PRINT "MEAN= ";ME%(N,RN); " STD.DEV.= ";ST%(N,RN); " RSD%=
"; LEFT$( RS$(N,RN),4)
2510 NEXT
2520 IF RT < 2 THEN INVERSE : PRINT " NOT ENOUGH STANDARDS TO
RUN DETECTION LIMITS!": NORMAL : GOTO 2640
2530 FOR RN = 1 TO RT
2540 FOR I = 1 TO NE
2550 C = C + C(I,RN)
2551 NEXT 2552 IF C = 0 THEN B = RN
2560 C = 0
2570 NEXT
2580 IF B = 0 THEN INVERSE : PRINT "NO BLANK HAS BEEN RUN!":
NORMAL : GOTO 2640
2590 FOR RN = 1 TO RT
2591 FOR N = 1 TO NE
2592 IF C(N,RN) = 0 GOTO 2630
2600 IF ME(N,RN) - ME(N,B) = 0 THEN D(N,RN) = 0: GOTO 2630
2610 D(N,RN) = (2000 * C(N,RN) * ST(N,B)) / (ME(N,RN) - ME(N,B))

```

```

2620 D$(N,RN) = STR$(D(N,RN))
2630 NEXT : NEXT
2640 PRINT CHR$(7): PRINT "DO YOU HAVE MORE STANDARDS TO RUN?":
GET B$: IF B$ < > "N" GOTO 2120
2650 IF RT = 0 THEN INVERSE : PRINT "NO STANDARDS HAVE BEEN
RUN!!": NORMAL : GOTO 3130
2660 IF A = 1 GOTO 3130
2670 IF A = 2 THEN PRINT D$: PRINT D$;"PR#1"
2680 ON H GOSUB 3370,3370,2780,3370,3370,2900,3370,3370,2780,3710
2710 ON H GOSUB 2780,2780,2900,2900,2900,2900,2780,2780,2900,3710
2740 ON H GOTO 3090,3090,3090,2960,2960,2960,3090,3090,3090,3710
2750 REM *****
2760 REM *DATA PRINT*
2770 REM *****
2780 FOR N = 1 TO NE
2790 FOR RN = 1 TO RT
2800 PRINT CHR$(18)"ELEMENT "; CHR$(14);EL$(N); CHR$(20);"
RUN=";RN;" REPS.=";R(RN);" CONC.=";C(N,RN);" UG/ML";
2810 IF ME(N,RN) > 32767 THEN PRINT " MEAN=";ME(N,RN);"
STD.DEV.=";ST$(N,RN);" RSD%="; LEFT$(R$(N,RN),4): GOTO 2830
2820 PRINT " MEAN=";ME$(N,RN);" STD.DEV.=";ST$(N,RN);" RSD%=";
LEFT$(R$(N,RN),4)
2830 IF R$(RN) < > "" THEN PRINT R$(RN)
2840 IF C(N,RN) < > 0 THEN PRINT CHR$(14);"DET. LMT. "; LEFT$(
D$(N,RN),6);" UG/L "; CHR$(20);
2850 PRINT CHR$(15);"RAW DATA ";
2860 FOR I = 1 TO R(RN): PRINT EL(N,RN,I);" ";
2870 NEXT
2880 PRINT : NEXT : PRINT : NEXT
2890 PRINT CHR$(18)
2900 RETURN
2910 REM *****
2920 REM *END OF DATA PRINT*
2930 REM *****
2940 REM *DETECTION LIMIT PRINT*
2950 REM *****
2960 IF RT < 2 THEN INVERSE : PRINT " NOT ENOUGH STANDARDS TO
RUN DETECTION LIMITS!": NORMAL : GOTO 3130
2970 FOR RN = 1 TO RT
2980 FOR I = 1 TO NE
2990 C = C + C(I,RN)
2991 NEXT
2992 IF C = 0 THEN B = RN
3000 C = 0
3010 NEXT
3020 IF B = 0 THEN INVERSE : PRINT "NO BLANK HAS BEEN RUN!":
NORMAL : GOTO 3130
3030 FOR RN = 1 TO RT
3031 FOR N = 1 TO NE
3032 IF C(N,RN) = 0 GOTO 3080
3040 IF ME(N,RN) - ME(N,B) = 0 THEN D(N,RN) = 0: GOTO 3060
3050 D(N,RN) = (2000 * C(N,RN) * ST(N,B)) / (ME(N,RN) - ME(N,B))
3060 D$(N,RN) = STR$(D(N,RN))

```

```

3070 IF C(N,RN) < > 0 THEN PRINT CHR$(18);"DET. LMT. AT
CONC.=";C(N,RN)" FOR "; CHR$(14);EL$(N); CHR$(20);" IS "; LEFT$(
D$(N,RN),6);" UG/L ";R$(RN)
3080 NEXT : NEXT
3090 PRINT D$;"PR#0"
3100 REM *****
3110 REM *END DETECTION LIMITS*
3120 REM *****
3130 PRINT : HTAB 8: PRINT "DO YOU WANT TO CONTINUE": PRINT
"USING THE SAME ELEMENTS AND SAME DATA?": GET B$
3140 IF B$ = "Y" THEN PRINT "DO YOU WANT TO CHANGE PRINT
INSTRUCTIONS? ": GET X$: IF X$ = "Y" THEN GOTO 240
3150 IF B$ = "Y" GOTO 2640
3160 PRINT : PRINT : PRINT "DO YOU WANT TO RUN NEW DATA?": GET
B$: IF B$ < > "Y" GOTO 3190
3170 IF A < > 2 THEN PRINT : CLEAR : GOTO 170
3180 IF B$ = "Y" THEN PRINT D$: PRINT D$;"PR#1": PRINT CHR$(
12): PRINT D$;"PR#0": CLEAR : GOTO 170
3190 PRINT : PRINT : PRINT "DO YOU WANT TO QUIT?": GET B$: IF B$
< > "Y" THEN GOTO 3220
3200 IF A < > 2 THEN PRINT : HOME : PRINT D$;"PR#0": PRINT
"ATOMIC FLUORESCENCE PROGRAM ENDED!": PRINT : PRINT " ENTER RUN TO
BEGIN AGAIN": END
3210 IF B$ = "Y" THEN PRINT D$: PRINT D$;"PR#1": PRINT CHR$(
12): HOME : PRINT D$;"PR#0": PRINT "ATOMIC FLUORESCENCE PROGRAM
ENDED!": PRINT : PRINT " ENTER RUN TO BEGIN AGAIN": END
3220 PRINT D$;"PR#1": PRINT : PRINT : PRINT : PRINT : PRINT :
PRINT D$;"PR#0": GOTO 3130
3230 REM *****
3240 REM **KEYBOARD INPUT**
3250 REM *****
3260 FOR N = 1 TO NE
3270 FOR I = 1 TO R(RN)
3280 PRINT : PRINT "FOR ELEMENT ";EL$(N);" CONC. ";C(N,RN);"
READING= ";
3290 INPUT EL(N,RN,I)
3300 NEXT
3310 NEXT
3320 FOR N = 1 TO NE
3321 FOR I = 1 TO R(RN)
3322 GOTO 2400
3330 REM **END KEYBOARD IN**
3340 REM *****
3350 REM *** HEADER PRINT ***
3360 REM *****
3370 IF TIS < > "" THEN X = LEN (TIS):I = 20 - X / 2: PRINT
TAB( I): PRINT CHR$(14);TIS; CHR$(20): PRINT
3380 IF OPS < > "" THEN PRINT "OPERATOR ";OPS;
3390 PRINT TAB( 41);"POSITION ELEMENT CURRENT PMT"
3400 IF (DAS < > "") OR (MOS < > "") OR (YRS < > "") THEN
PRINT "DATE ";DAS;" ";MOS;" ";YRS;
3410 IF TS < > "" THEN PRINT " TIME ";TS;
3420 IF MP$(1) < > "" THEN PRINT TAB( 45);MP$(1); TAB(
14);EL$(1); TAB( 23);LC$(1); TAB( 31);PM$(1)

```

```

3030 IF POS < > "" THEN PRINT "POWER ";POS;: IF MP$(2) = ""
THEN PRINT
3440 IF MP$(2) < > "" THEN PRINT TAB( 45);MP$(2); TAB(
14);EL$(2); TAB( 23);LC$(2); TAB( 31);PM$(2)
3450 IF CAS < > "" THEN PRINT "CARRIER FLOW ";CAS;: IF MP$(3)
= "" THEN PRINT
3460 IF MP$(3) < > "" THEN PRINT TAB( 45);MP$(3); TAB(
14);EL$(3); TAB( 23);LC$(3); TAB( 31);PM$(3)
3470 IF COS < > "" THEN PRINT "COOLANT FLOW ";COS;: IF MP$(4)
= "" THEN PRINT
3480 IF MP$(4) < > "" THEN PRINT TAB( 45);MP$(4); TAB(
14);EL$(4); TAB( 23);LC$(4); TAB( 31);PM$(4)
3490 IF PUS < > "" THEN PRINT "PUMP SPEED ";PUS;: IF MP$(5) =
"" THEN PRINT
3500 IF MP$(5) < > "" THEN PRINT TAB( 45);MP$(5); TAB(
14);EL$(5); TAB( 23);LC$(5); TAB( 31);PM$(5)
3510 IF OXS < > "" THEN PRINT "OXIDANT FLOW ";OX$;: IF MP$(6)
= "" THEN PRINT
3520 IF MP$(6) < > "" THEN PRINT TAB( 45);MP$(6); TAB(
14);EL$(6); TAB( 23);LC$(6); TAB( 31);PM$(6)
3530 IF RES < > "" THEN PRINT "REDUCTA#^ FLOW ";RES;: IF
MP$(7) = "" THEN PRINT
3540 IF MP$(7) < > "" THEN PRINT TAB( 45);MP$(7); TAB(
14);EL$(7); TAB( 23);LC$(7); TAB( 31);PM$(7)
3550 IF MP$(8) < > "" THEN PRINT TAB( 45);MP$(8); TAB(
14);EL$(8); TAB( 23);LC$(8); TAB( 31);PM$(8)
3560 IF MP$(9) < > "" THEN PRINT TAB( 45);MP$(9); TAB(
14);EL$(9); TAB( 23);LC$(9); TAB( 31);PM$(9)
3570 IF MP$(10) < > "" THEN PRINT TAB( 45);MP$(10); TAB(
14);EL$(10); TAB( 23);LC$(10); TAB( 31);PM$(10)
3580 IF MP$(11) < > "" THEN PRINT TAB( 45);MP$(11); TAB(
14);EL$(11); TAB( 23);LC$(11); TAB( 31);PM$(11)
3590 IF MP$(12) < > "" THEN PRINT TAB( 45);MP$(12); TAB(
14);EL$(12); TAB( 23);LC$(12); TAB( 31);PM$(12)
3600 IF R1$ < > "" THEN PRINT "REMARKS ";R1$
3610 IF R2$ < > "" THEN PRINT "REMARKS ";R2$
3620 IF R3$ < > "" THEN PRINT "REMARKS ";R3$
3630 IF R4$ < > "" THEN PRINT "REMARKS ";R4$
3640 PRINT : PRINT D$;"PR#0"
3650 RETURN
3660 REM *****
3670 REM END HEADER PRINT**
3680 REM *****
3690 REM MENU MODE PRINT
3700 REM *****
3710 HOME
3720 INVERSE : PRINT "SELECT EXTERNAL PRINTER IN MENU MODE":
PRINT : PRINT "HIT CONTROL RESET TO ESCAPE!": NORMAL
3730 PRINT D$: PRINT D$;"IN#2"
3740 FOR I = 0 TO 71
3750 GET A$(I)
3760 NEXT 3770 FOR I = 72 TO 92
3780 INPUT A$(I)
3790 IF RIGHT$(A$(I),6) = "CYCLES" GOTO 3810
3800 NEXT

```

```
3810 PRINT D$;"IN#0"
3820 PRINT D$: PRINT D$;"PR#1"
3030 FOR I = 0 TO 71
3840 PRINT A$(I);
3850 NEXT : PRINT
3860 FOR I = 72 TO 92
3870 PRINT A$(I)
3880 IF RIGHT$(A$(I),6) = "CYCLES" GOTO 3900
3890 NEXT
3900 PRINT D$;"PR#0"
3910 GOTO 3720
3920 REM *****
3930 REM END OF MENU PRINT
3940 REM *****
```

PROGRAM OUTPUT SECTION

The following is an example of a printout produced using the option of all data full header.

DETECTION LIMITS PROGRAM EXAMPLE

OPERATOR ALLEN MEIER	POSITION	ELEMENT	CURRENT	FMT
DATE 30 JULY 1984 TIME AM	1	CA	18.2	600 V
POWER 550 WATTS	2	CU	13.4	650 V
CARRIER FLOW 1.3	4	K	10.3	850 V
COOLANT FLOW 12	6	ZN	11.1	700 V
PUMP SPEED 425	12	MG	18.2	600 V
OXIDANT FLOW 0				
REDUCTANT FLOW 50				

REMARKS THIS AN EXAMPLE OF A PRINTOUT!

REMARKS THE VALUES LISTED HERE ARE ONLY FOR EXAMPLE!

ELEMENT **CA** RUN=1 REPS.=10 CONC.=0 UG/ML MEAN=1786 STD.DEV.=33 RSD%=1.86
 RAW DATA 1777 1767 1787 1776 1877 1776 1787 1787 1756 1775 NET COUNTS= 0
 ELEMENT **CA** RUN=2 REPS.=3 CONC.=10 UG/ML MEAN=5595 STD.DEV.=61 RSD%=1.10
DET. LMT. 174.53 UG/L RAW DATA 5555 5565 5667 NET COUNTS= 3809.16667

ELEMENT **CU** RUN=1 REPS.=10 CONC.=0 UG/ML MEAN=739 STD.DEV.=369 RSD%=49.9
 RAW DATA 1787 606 621 634 585 598 578 668 657 665 NET COUNTS= 0
 ELEMENT **CU** RUN=2 REPS.=3 CONC.=1 UG/ML MEAN=9880 STD.DEV.=6 RSD%=.067
DET. LMT. 80.814 UG/L RAW DATA 9876 9877 9888 NET COUNTS= 9140.43333

ELEMENT **K** RUN=1 REPS.=10 CONC.=0 UG/ML MEAN=59 STD.DEV.=189 RSD%=316.
 RAW DATA 599 0 0 0 0 0 0 0 0 0 NET COUNTS= 0
 ELEMENT **K** RUN=2 REPS.=3 CONC.=.5 UG/ML MEAN=3010 STD.DEV.=9 RSD%=.332
DET. LMT. 64.208 UG/L RAW DATA 3000 3020 3010 NET COUNTS= 2950.1

ELEMENT **ZN** RUN=1 REPS.=10 CONC.=0 UG/ML MEAN=5 STD.DEV.=4 RSD%=70.2
 RAW DATA 0 9 9 0 8 9 0 9 7 7 NET COUNTS= 0
 ELEMENT **ZN** RUN=2 REPS.=3 CONC.=2 UG/ML MEAN=10100 STD.DEV.=100 RSD%=.990
DET. LMT. 1.6155 UG/L RAW DATA 10000 10100 10200 NET COUNTS= 10094.2

ELEMENT **MG** RUN=1 REPS.=10 CONC.=0 UG/ML MEAN=123 STD.DEV.=309 RSD%=250.
 RAW DATA 7 8 8 8 9 90 90 999 9 9 NET COUNTS= 0
 ELEMENT **MG** RUN=2 REPS.=3 CONC.=5 UG/ML MEAN=20066 STD.DEV.=57 RSD%=.287
DET. LMT. 155.15 UG/L RAW DATA 20000 20100 20100 NET COUNTS= 19942.9667